
News 
 
Version 5.05 of Spike2 is now available from www.ced.co.uk/upu.shtml for registered v5 users. This 
revision includes interactive curve fitting to time, result and XY views and creation of new data 
channels based on active cursor measurements. You can read the full list of new features on our web 
site. 
 
A demonstration version of the current software is also available from www.ced.co.uk/pru.shtml 
 
Version 4.18 of Spike2 is downloadable for registered v4 users. 
Version 2.14 of Signal is also downloadable for registered v2 users.  
 
Our U.K training days will take place on the 25th – 26th of March 2004 in Cambridge. Details of the 
training day sessions and a registration form are available from: www.ced.co.uk/nw9u.htm. 
 
We will also be holding additional US training days on the 15th – 16th April 2004, prior to the 
Experimental Biology meeting in Washington DC. Details of the training day sessions and a 
registration form are available from: www.ced.co.uk/nw9y.htm. 
 
The Spike2 and Signal scripts for this edition of the newsletter have been included as embedded 
attachments in the document. To load the scripts, simply double-click on the appropriate icon beside 
the script file name. If you have Spike2 or Signal running on your PC, the appropriate script will load in 
a normal script file window. Alternatively, you can right click on the icon and save the embedded file to 
disk.  
 
If you have any problems accessing the embedded files, please let us know via the return email 
address at the bottom of the newsletter.  
  
 
SPIKE2 
 
Q. I am recording Blood Pressure data and would like an on-line display of heart rate from this 

trace. 
 
A. There is a new Analysis menu measurements command in Spike2 version 5.05 that creates a 

new data channel based on active cursor measurements. This function can be used on-line to 
generate a heart rate display from an ECG or blood pressure channel in real time during 
sampling. Previously a script was required to display on-line heart rate in this way. 

 

#6 ~ 2nd February 2004 

http://www.ced.co.uk/s2wu5u.shtml
http://www.ced.co.uk/s2demu.shtml
http://www.ced.co.uk/nw9u.htm
http://www.ced.co.uk/nw9y.htm


 Open a Spike2 data file ready for sampling. From the Analysis menu, select Measurements 
and then Data channel to open the dialog. The dialog has a number of fields, some of which 
are common to those in the Measurements to XY views dialog. 

   

  
  

Select the new channel to create from the Target data channel field. This can be a new 
memory channel or one of the unused disk channels. The Channel type field holds a list of 
channel types that can be created; Event, Marker or RealMark. If you select RealMark, the 
target channel has one real value (e.g. Amplitude) attached to each marker; this value is set 
by the Y measurements field. As we only want to mark peaks in this case, with no associated 
Y data value, we can select an Event channel to mark the peak times. 
 

 The fields in the Cursor 0 stepping dialog are identical to those in the Measurements to XY 
view dialog. This section is used to set the search method for measurements. Set the method 
to peak find, and set the amplitude to find the peaks in the blood pressure trace.  Spike2 
detects peaks by monitoring the waveform level to ensure that the signal falls by at least the 
amplitude set here. For a BP channel calibrated in mmHg, 25 would be a suitable estimate for 
our example.  

 

  



To avoid marking any dicrotic notch present in the BP data, we can also set a minimum step 
for the cursor.   

 
 The X measurements section sets the time of each event or marker that is added into the new 

data channel. This would normally be an expression related to cursor positions, in this case, 
the time of each peak as found by Cursor(0). 

 
 The Y measurements section is only available when the type of the target channel is 

RealMark. The value measured here sets the RealMark data value.  
 
 Pressing New then opens the on-line processing dialog.  

 

 
       

From the process dialog we can select the update mode for the measurements. For this 
example, we can use Automatic, Update window every 1 second and click OK. Now if you 
start sampling you should see on-line marking of the BP peaks as events. If you find that no 
peaks are being marked, adjust the Peak Find amplitude settings in the Measure to channel 
dialog. The process settings for the measurements can be changed at any time by right-
clicking on the channel and selecting Process settings from the menu. 
 
Finally, select Channel Draw mode from the view menu and choose instantaneous frequency 
from the drop-down list. Check the Per minute box to display as beats per minute.  

     
 
SIGNAL 
 
  We had a number of requests at the recent US Neuroscience meeting for new analysis and 

display features to be implemented in Signal. We welcome any suggestions from current 
Signal users on features that they feel may improve the software. Please let us know if you 
have a specific request that you would like to see included in a future release of Signal. 

     
 
Scripts. Spike2       (Right mouse click icons and save embedded script file[s] to disk) 
 
Q.  I would like to use Spike2 to output pulses that ramp linearly from 1Hz to 30Hz and then back 

to 1Hz, over pre-defined time durations during sampling. Spike2 should take into account the 
reduced time delay between each output pulse over the specified time range to ensure a 
linear ramp of output pulses, rather than an exponential curve.  

 
A.  The attached LinRampFreq.s2s script and associated LinRampFreq.pls 

sequence creates ramps of increasing or decreasing pulse frequency on-line. The script 
allows the user to enter settings for ramp time, as well as start, turning point and end 
frequencies for the pulse output. The user can also select whether to produce a single 
increasing or decreasing ramp. Before running the script, you must ensure that the 


'LinRampFreq.s2s|Script to create increasing or decreasing pulse frequency ramp
'User can specify start, turning point and end frequencies as well as ramp time
'Hides all unwanted windows and restores them.
'Manages the data files (keeps one file at a time).
'
'Note this script is UNDER DEVELOPMENT BY CED and is included for
'illustration only. We make no warranties as to its suitability
'or otherwise. It has not been thoroughly tested.
'You should use it 'as is'
'
var data%;								'Handle of new data file (Spike2 sets it to 0)
var sTime;								'The last time we looked at the idle routine
var startFreq := 1;					'Frequency at start of up ramp
var maxFreq := 30;					'Maximum frequency
var endFreq := 1;						'Frequency at end of down ramp
var doUpFlag% := 1;					'Flag to do upward ramp
var doDownFlag% := 1;				'  "    "   downward 
var upRTime := 10;					'Time of up ramp
var downRTime := 10;					'  "    down
var numEvents%;						'Number of events in ramp sequence
var rampFlag% := 0;					'Flag if ramp in progress
var stimChan% := 1;					'Event channel recording stimuli
var prevStims%;
var numStims%;							'Number of events in channel
var int%;								'Array element of current interval	

var freq[10000];						'Array of frequencies
var int[10000];						'Array of intervals
var tInt[10000];						'Temporary array of intervals
var stimTime;							'Time of stimulus relative to start of ramp

var ticks%;
var evFlag%;

HideAll();                       'Hide all unwanted bars
ToolbarVisible(1);					'Make toolbar visible always
New%();									'Set up new sampling window
DoToolbar();							'Do the toolbar
RestoreAll();                    'Restore all bars
Halt;

func New%()								'New sampling window
View(LogHandle());					'Make log view the current view
EditSelectAll();						'Select all text in log view
EditClear();							'Delete it
Window(0,80,100,100);				'Display it at the bottom of the screen
WindowVisible(1);						'Make it visbible

if data%>0 then						'If there is already a data view open...
  View(data%);							'...close it
  FileClose();                   'No point setting data% to 0...
endif;

data%:=FileNew(0,3);					'Open a new data file for sampling
if data%<0 then Message("Unable to open new data file");Halt() endif;
Window(0,0,100,80);					'Make data window in top bit of screen
XRange(0,10);

FrontView(LogHandle());				'Bring the Log view to the front
FrontView(data%);						'Bring the data view to the front
ToolbarEnable(3,0);					'Disable "Sample stop" button
ToolbarEnable(2,1);					'Disable "Sample stop" button
ToolbarText("Press SAMPLE START to commence sampling");
return 1;
end;

proc DoToolbar()
ToolbarSet(1,"&Quit",Quit%);		'Set up toolbar buttons
ToolbarSet(2,"&Sample start", Start%);
ToolbarSet(3,"&Sample stop", Stop%);
ToolbarSet(4,"&New file", New%);
ToolbarSet(6, "Set pulses", SetPulses%);
ToolbarSet(8, "Pulses", DoPulses%);
ToolbarEnable(3,0);					'Disable "Sample stop" button
ToolbarEnable(5,0);
Toolbar("Press SAMPLE START to commence sampling", 1023);'Wait here until quit
end;

func Quit%()							'If "Quit" is pressed
SampleStop();							'Stop sampling
return 0;								'leave toolbar
end;

func Start%()							'If "Start" is pressed
SampleStart();							'Start sampling
prevStims% := 0;						'Set number of stimuli to zero
ToolbarEnable(4,0);					'Disable "New file" button
ToolbarEnable(3,1);					'Enable "Sample stop" button
ToolbarEnable(2,0);					'Disable "Sample start" button
ToolbarEnable(1,0);					'Disable "Quit" button
ToolbarEnable(5,1);
ToolbarText("Press SAMPLE STOP to stop sampling");
return 1;								'Stay with toolbar
end;

func Stop%()							'If "Stop" is pressed
SampleStop();							'Stop sampling
if (data%>0) and                 'if no data in file, stop will close it
   (ViewKind(data%) <> 0) then data% := 0 endif;
ToolbarEnable(4,1);					'Enable "New file" button
ToolbarEnable(3,0);					'Disable "Sample stop" button
ToolbarEnable(1,1);					'Enable "Quit" button
ToolbarText("Press FILE NEW to capture more data");
return 1;								'Stay in toolbar
end;

Func SetPulses%()
var sectionEvents% := 0;
numEvents% := 0;
ArrConst(int[],0);
DlgCreate("Frequency ramp setup");
DlgText("Ramp 1",23,1);
DlgText("Ramp 2",37,1);
DlgText("Frequencies",2,2);
DlgText("Initial",17,2);
DlgText("Turning Point",28,2);
DlgText("End",45,2);
DlgReal(1,4,0.1,1000,18,3);
DlgReal(2,4,0.1,1000,32,3);
DlgReal(3,4,0.1,1000,46,3);
DlgText("Times",2,4);
DlgReal(4,4,1,1000,25,4);
DlgReal(5,4,1,1000,39,4);
DlgCheck(6,"",25,2);
DlgCheck(7,"",40,2);
DlgShow(startFreq,maxFreq,endFreq,upRTime,downRTime,doUpFlag%,doDownFlag%);
if doUpFlag% then
	sectionEvents% := GetIntArray(startFreq,maxFreq,upRTime);
	ArrConst(int[:sectionEvents%],tInt[:sectionEvents%]);
	numEvents% += sectionEvents%;
endif;
if doDownFlag% then
	sectionEvents% := GetIntArray(maxFreq,endFreq,downRTime);
	ArrConst(int[numEvents%:sectionEvents%],tInt[:sectionEvents%]);
	numEvents% += sectionEvents%;
endif;
return 1;
end;

Func DoPulses%()
View(data%);
Seconds(0);
int% := 0;
ticks% := Trunc((int[int%])*10000)-2;
SampleSeqVar(1,ticks%);
SampleKey("g");
ToolbarSet(0,"",Idle%);				'Call Idle%() whenever there is free time
return 1;
end;

func Idle%()							'Idle routine runs when system has time
var eTime;
if ViewKind(data%)=0 then
   View(data%);
   eTime:=Maxtime();
	if int% <= numEvents% then
		if Count(stimChan%,0,eTime) > prevStims% then
			int% += 1;
			ticks% := Trunc((int[int%])*10000)-2;
			SampleSeqVar(1,ticks%);
			prevStims% += 1;
		endif;
	endif;
endif;
return 1;								'Stay in toolbar
end;

'You can find this code in the App() on-line help.
var gFloat%[20];              'global for floating window states
proc HideAll()
var i%;
gFloat%[0] := App(-2);        'number of windows

for i% := 1 to gFloat%[0] do  'hide all windows and save state
   gFloat%[i%] := View(App(i%)).WindowVisible(0);
   next;
end

proc RestoreAll()
var i%;
for i% := 1 to gFloat%[0] do  'restore hidden windows
   View(App(i%)).WindowVisible(gFloat%[i%]);
   next;
end


Func GetIntArray(f0,f1,T);
var k,fn;
var i%;
var sqr;
var stimTime;
k := (f1-f0)/(T-(1/f0));
freq[0] := f0;
tInt[0] := 1/freq[0];
repeat
 	fn := freq[i%]/2;
	sqr := fn*fn + k;
	if sqr >0 then
	 	i% += 1;
		freq[i%] := fn + Sqrt(fn*fn + k);
		tInt[i%] := 1/freq[i%];
		stimTime := ArrSum(tInt[:i%+1]);
	else
		stimTime := T;
	endif;
until stimTime >= T;
return i% + 1;
end;



CED Ltd
LinRampFreq.s2s
LinRampFreq.s2s




CED Ltd
LinRampFreq.pls
LinRampFreq.pls



LinRampFreq.pls output sequence is loaded and made the current sequence. The script 
sets up a new data file for recording based on the current sampling configuration. Set channel 
1 in the sampling configuration to be an Event channel in which pulses can be recorded to 
help visualize the results. The script requires this data to be recorded, as it keeps a count of 
the number of pulses produced. The pulses are currently set to be 10 microseconds long, and 
are output from digital output 0 on the 1401. 

 
This script has been tested and is known to work in Spike2 v4 and v5.  

 
 
Scripts. Signal  
 
Q.  I would like to be able to quickly view areas of interest in individual frames; perhaps by 

showing sections of data between cursors in a duplicate window in Signal.  
 
A.  The attached CursorView.sgs  script creates a duplicate window of an open time 

view, and continuously updates the view in the duplicate to show data between cursors. 
 
 This script is for Signal v2.  (Right mouse click icon and save embedded script file to disk)

 
 
Did you know…? 
 
You can copy individual frames from one data file to another in Signal. Use the Edit menu Copy 
command to copy a single frame of data from a data file to the clipboard. If you then open another 
data file and select the Append frame function from the Analysis menu, you can paste the frame data 
from the clipboard into this new frame. 
 
 
Recent questions 
 
Q.  I am using Spike2 to template synchronised spikes from two waveform channels off-line into 

two separate WaveMark channels. There are occasions when there is a spike on one channel, 
but not on the other. I would like to record this non-spiking data to disk to prove that this 
synchronisation is not absolute, but cannot see a way to do this using the WaveMark method, 
as events are only detected by threshold crossings.  

 
A. Using the n-trode analysis feature in Spike2 version 5, we can generate a stereotrode trace 

from the two existing waveform channels. This will save the spike data from both channels into 
a single WaveMark channel. Simply select the two channels by holding Ctrl and clicking on the 
channel numbers, and select New n-trode from the Analysis pull-down menu. When using the 
spike template functions, spikes are detected by the signal crossing a trigger level. This 
causes the data around the trigger point to be stored as a WaveMark. With multiple traces, the 
template window is split horizontally to show an oscilloscope view of the data from each 
selected channel. The horizontal trigger levels in these windows are independent and can be 
set to different levels depending on the input data. Each trace is tracked separately and the 
first trace to trigger saves the corresponding data for all traces. If a spike on any one channel 
crosses the trigger level, it will also save the corresponding time range from all other selected 
channels.  

 
 
User group 
 
We are in the process of setting up an archive message board on our web site. As soon as this is 
done we will use this newsletter to announce that it is available. This seems to be the most sensible 
route to go rather than direct emailing both for security and for minimising spam. 
 
If you wish to un-subscribe from the mailing list for this newsletter, please notify sales@ced.co.uk. 

mailto:sales@ced.co.uk

'$CursorView|Script to set time displayed in a duplicate window based on cursor positions in the original data view
var file%;							'Data file handle
var dup%;							'Duplicate window handle
var sTime,eTime;					'Start and end of display section
var presTime,preeTime;				'Current start and end time of duplicate display	
var frame%,frameDup%;				'Frame numbers of data and duplicate windows

ToolbarSet(0, "Cursor view",MoveView%);		'Link to idle routine to track changes
ToolbarSet(1, "Quit");						'Toolbar buttons
ToolbarSet(3, "Open File",Open%);
Toolbar("Window to show area between cursors",1023);


Func Open%()						'Function to open a data file
file%:=FileOpen("",0,0);			'Open data file	
Window(0,0,50,100);					'Position window in left half of screen
CursorSet(2);						'Set 2 cursors
CursorLabel(4,1,"Start");			'Label cursors
CursorLabel(4,2,"End");
FrontView(file%);
dup%:=WindowDuplicate();			'Duplicate window
View(dup%).Window(50,0,100,100);	'Position duplicate in right half of screen
CursorDelete();						'Delete cursors in duplicate view		
CursorDelete();
FrontView(dup%);
Return 1;
End;

Func MoveView%()					'Idle routine 
if dup%>0 then						'Do nothing unless duplicate window is open 
	View(file%);					'Ensure data window is current view
	sTime:=Cursor(1);				'Get times of cursors
	eTime:=Cursor(2);
	if eTime<sTime then				'Ensure cursors are in correct order
		CursorRenumber();
		CursorLabel(4,1,"Start");			'Label cursors
		CursorLabel(4,2,"End");
	endif;
	frame%:=Frame();				'Get frame number of data window
	frameDup%:=View(dup%).Frame();	'Get frame number of duplicate window
	if frameDup% <> frame% then		'If frames are different	
		View(dup%).Frame(frame%);	'Change duplicate window to match data window
	endif;
	docase									
	case sTime<>presTime then				'If start time has changed
		View(dup%).XRange(sTime,eTime);		'Adjust start time in duplicate
	case eTime<>preeTime then				'Same with end times
		View(dup%).XRange(sTime,eTime);
	endcase;
	presTime:=sTime;						
	preeTime:=eTime;
endif;
Return 1;
End;

